
Abstract. This is a preliminary work on configuration knowledge representation which serves as
a foundation for building interactive configuration systems in Answer Set Programming (ASP).
The major concepts of the product configuration problem are identified and discussed with a bike
configuration example. A fact format is developed for expressing product knowledge that is domain-
specific and can be mapped from other systems. Finally, a domain-independent ASP encoding is
provided that represents the concepts in the configuration problem.

1 Introduction

Product configuration is one of the most successful commercial applications of artificial intelligence
techniques [19] [3]. It has been used from telephone switching systems [22], to smart home configurations
[4], in cement manufacturing plants [18], and the automotive industry [23]. Being highly dynamic, this
field evolved from mass production to mass customisation [24], as the customer needs became more
individualistic. The recent trend is interactive configuration [1], where the user is involved in every step of
the configuration process.

ASP [9] is a declarative modeling approach used for product configuration. In fact, product configura-
tion was one of the first practical applications of ASP [20]. Since then, there have been many works that
use ASP to represent product configuration [8] [17] [4]. But most of them are either domain-specific or
tackle a subset of the configuration problem. Others express product knowledge in form of ASP rules.
This requires domain-experts to be proficient in ASP, which is not always the case.

This work serves as a foundation for interactive configuration systems in ASP. The motivation
here is the identification of the major concepts related to product configuration and the corresponding
representation in ASP. The development of a domain-independent configuration knowledge representation
is another motivation. This can be achieved by a clear separation between product knowledge and
configuration knowledge. The advantage here is that the configuration knowledge can be reused for any
product to be configured and ensures better system maintenance. Stress is given for representing product
knowledge as facts, to facilitate seamless mapping from other systems. This enables domain-experts to
provide product knowledge without being experts in ASP [25].

This paper is organized as follows: In Section 2, a background on product configuration is given. The
configuration problem is described in Section 3 using bike configuration as an example. An ASP-based
representation of the configuration problem and the corresponding solution is given in Section 4. Finally,
the future work is expressed in Section 5.

2 Background

One of the earliest applications of product configuration was R1/XCON, built on a rule-based repre-
sentation language, where the solving knowledge was intertwined with domain-specific configuration
knowledge [14]. The trend quickly evolved to model-based knowledge representation where there is a

http://dx.doi.org/10.4204/EPTCS.345.46
https://creativecommons.org
https://creativecommons.org/licenses/by/4.0/


S. Mishra 297

clear separation of problem solving knowledge from domain-specific configuration knowledge. As the
industry shifted from mass production to mass customisation, another boundary between domain-specific
configuration knowledge and user requirements was drawn [16].

To create a general understanding of configuration, [21] introduced concepts like component types,
properties and constraints like partof that are discussed in detail in the Section 3. Also, the configuration
is classified to connection-based, structure-based, resource-based, and function-based approaches. Most
concepts used in this work is adapted from [21] and [11].

Representation and modeling of configuration knowledge have been split into three categories, namely
constraint-based, graphical and logic-based knowledge representation [11]. Constraint programming
has been one of the major used techniques for product configuration. Starting with static constraint
satisfaction where all the variables had to be assigned values, the switch was done to dynamic constraint
satisfaction [15], where the configuration variables can be activated and deactivated during the search
process. Owing to the lack of support for component-oriented modelling in the previous techniques,
generative constraint satisfaction (GCSP) [22] emerged. GCSP was designed for product configuration
and has widespread uses in the industry.

The main graphic-based representation systems used for constraint satisfaction are feature models and
UML component diagrams [13], [5]. These systems improve the accessibility of configuration knowledge
to domain experts who are not acquainted with modeling, since these systems are more understandable
and maintainable. Predicate, as well as description logic, have been used for representing configuration
problems, along with a mapping from UML diagrams [5].

ASP has been used widely in product configuration [20] [23]. Linux Package Configuration in [8] is
another interesting application of ASP. An ASP representation of a configuration problem in the context
of feature models is given in [17]. Mapping of object-oriented concepts to ASP directed at product
configuration in UML was shown in [2]. In [4], an introductory example of representing smart home
configurations in ASP is provided.

3 Product Configuration Problem

In this section, some of the major concepts used in product configuration are described in the context of
bike configuration. The building blocks of a configuration problem are components. The components in
the bike configuration problem consist of a bike, a frame, two wheels namely the front wheel and the rear
wheel, a stand, and a basket as shown in Figure 1. The domain of each component in the configuration
problem is a set of component type, shortened to type. Examples of types include w1, w2, and w3 which
are in the domain of both the front wheel and the rear wheel.

A component is characterized by a set of properties. The types also have properties, whose values are
predefined. For example, the material of frame type f1 is aluminum. The domain of component properties
apart from type is given by the predefined property values of the corresponding types of components. As
an example in the bike configuration example, the material of the frame can be either carbon fiber or
aluminum. To make the representation more uniform, type is considered as a property of the component.

A configuration problem can be defined as a set of components with properties, a set of domains for
each component property, and a set of constraints listed below:

• Property Assignment: These constraints deal with the assignment of values to component proper-
ties from their respective domains.

– A1: Every component present in the configuration solution must be assigned a type.
– A2: A component property can only have one value.

– A3: The values assigned to a component property must exactly correspond to the predefined
property values of the type it is assigned.

– A4: All mandatory properties of a component should be assigned values if the component is
in a configuration solution.

• Partonomy: The structure of a product can be represented as a partonomy where a whole component
may have optional and mandatory parts. For example, a bike is a whole component while a frame is
a mandatory part and a basket is an optional part of the bike.

– P1: If a part component is present in an assignment, the whole component must be also in the
configuration.

– P2: If a component is present in an assignment, all the mandatory part components have to be
in the assignment.

• Requirements: These constraints specify that including certain components (with property values)
forces other components (with property values) to be present in the configuration.

– R1: A component requires another component to be part of the solution.
– R2: To satisfy the configuration, a component requires another component with a specific

property value to be assigned to the configuration (or vice versa).
– R3: To satisfy the configuration, a component with a certain property value requires another

component with a specific property value to be assigned to the configuration.

• Incompatibility: These constraints specify that certain combinations of components (with property
values) are not allowed in the configuration solution.

– I1: Incompatible components cannot be together in a configured solution.
– I2: A component can be incompatible with a certain presence value of another component. In

this case, if the former component is present in the solution, then the later component can’t
have the respective property value.

– I3: The property value of a component can be incompatible with the property value of another
component. In this case, only one of the components and their property value can be part of
the solution or both have to be excluded.

• User Requirements: The users can specify their requirements in terms of components or specific
values of component properties.

– U1: Every component that the user requests, must be part of the configuration.
– U2: A user can require a component with a specific property value. In this case, the specified

component with the respective property value must be present in the solution.
– U3: Every component that the user requests not to be present, must not be part of the

configuration.
– U4: A user can require that a component with a specific property value is not in the configu-

ration. In this case, the specified component may be absent or may be present with another
value for the respective property in the solution.

A configuration solution is an assignment of values to component properties in the configuration
problem from their domains such that the constraints are satisfied.

Listing 1: Bike configuration problem instance as facts
1 domain(bike ,type ,city_bike ). domain(bike ,type ,mountain_bike ).

2 domain(rear_wheel ,type ,w1). domain(rear_wheel ,type ,w2).

3 domain(rear_wheel ,type ,w3). domain(front_wheel ,type ,w3).

4 domain(front_wheel ,type ,w1). domain(front_wheel ,type ,w2).

5 domain(frame ,type ,f1). domain(frame ,type ,f2). domain(frame ,type ,f3).

6 domain(stand ,type ,s1). domain(basket ,type ,b1).

7
8 property_val(f1,material ,aluminium ). property_val(f1,basket_support ,true).

9 property_val(f2,material ,carbon_fiber ). property_val(f2,basket_support ,false ).

10 property_val(f3,basket_support ,true). property_val(w1,material ,aluminium ).

11 property_val(w1,size ,28). property_val(w2,size ,26). property_val(w3,size ,28).

12 property_val(w2,material ,aluminium ). property_val(w3,material ,carbon_fiber ).

13
14 mandatory_property(frame ,material ).

15
16 partof(bike ,frame ,mandatory ). partof(bike ,front_wheel ,mandatory ).

17 partof(bike ,rear_wheel ,mandatory ). partof(bike ,basket ,optional ).

18 partof(bike ,stand ,optional ).

19
20 incompatible_com_pv(basket ,(bike ,type ,mountain_bike )).

21 incompatible_pv_pv (( front_wheel ,size ,26) ,( rear_wheel ,size ,28)).

22 incompatible_pv_pv (( front_wheel ,size ,28) ,( rear_wheel ,size ,26)).

23
24 require_com_com(basket ,stand).

25 require_com_pv(basket ,(frame ,basket_support ,true )).

26 require_pv_pv ((frame ,material ,aluminium),(front_wheel ,material ,aluminium )).

27
28 user_com(req ,bike). user_com(req ,basket ). user_com_pv(req ,( front_wheel ,size ,26)).

4 Representing Configuration in ASP

4.1 Fact Format

Listing 1 shows an instance of the bike configuration problem shown in Figure 1. The predicate
domain(C,P,V) expresses that some value V can be assigned to an property P of a component C.
Initially, domains of the type property of all components are provided in Lines 1-6. The predicate
property val(T,P,V) shown in Line 8-12 denotes that V is the predefined value of the property P of
type T. This predicate is also used to generate the domains of component properties apart from type during
preprocessing.

The predicate mandatory property(C, P) in Line 14 is used to express the mandatory property
constraint A4, where P is a mandatory property of a component C. During the preprocessing section of
the encoding, the property type is made a mandatory property for all components. Partonomy of the
bike is listed in Lines 16-18, where the predicate partof(C1,C2,V) expresses that a component C2 is a
mandatory or optional part of the component C1 depending on the respective value of V. The first fact
of Line 16 expresses that the frame is a mandatory part of the bike while the fact Line 18 expresses that
basket is an optional part of the bike.

Product specific incompatibility relations I1-I3 are represented in fact format in Line 20-22. The
predicate incompatible com pv(C1,(C2,P2,V2)), denotes an incompatibility relation I2 between a
component C1) and a component property value C2,P2,V2. For example, the fact in Line 20 expresses that
a mountain bike can’t have a basket. Incompatibility I3 between component attribute values (C1,P1,V1)
and (C2,P2,V2) in form of predicate incompatible pv pv((C1,P1,V1), (C2,P2,V2)) is shown in

Line 21-22, which specifies that the front wheel and rear wheel can’t be of different sizes. It can be noted
that incompatible com com(C1,C2) to express incompatibility between component C1 and C2 is not
given in this use case.

The requirements between bike components R1-R3 are given in Line 24-26. The condition R1
where component C1 requires C2 is expressed by predicate require com com(C1, C2). For exam-
ple, the fact in Line 24 expresses that a basket requires a stand. Similarly, require com pv(C1,

(C2,P2,V2)) expresses R2 where a component C1 requires another component C2 with property value
P2,V2. This example is shown in Line 25 expresses where a basket requires the frame to support a basket.
Requirement R3 of component attribute value (C2,P2,V2) by (C1,P1,V1) is in form of predicate
require pv pv((C1,P1,V1), (C2,P2,V2)). For example, in Line 26, the front wheel has to be made
up of aluminum if the frame is. Additionally, there is a predicate of form require com pv((C1,P1,V1),

C2), where presence of the component C1 with property value P1,V1 requires the component C2, which
hasn’t been used here.

Finally in Line 28, the user requirements are listed. Here user com(req, C) means that user requires
the component C. In the example, the user requires a bike and a basket. Users can also specify if they want
the property P of component C to have a value V, in form of the predicate user com(req, (C,P,V)).
For example, the user requires the size of the front wheel should be 26 is given in the last fact of Line 28.
The above two predicates express U1 and U2 respectively. U3 and U4 can be expressed in a similar way
by replacing req to nreq as the first arguement.

Listing 2: Configuration Problem Encoding

1 domain(C,P,V) :- domain(C,type ,T), property_val(T,P,V).

2 mandatory_property(C,type) :- domain(C,_,_).

3 require_com_com(Part ,Whole) :- partof(Whole ,Part ,_).

4 require_com_com(Whole ,Part) :- partof(Whole ,Part ,mandatory ).

5
6 {assign(C,P,V): domain(C,P,V)}.

7 component(C) :- assign(C,P,V).

8
9 :- assign(C,P,V1), assign(C,P,V2), V1 < V2.

10
11 :- component(C), mandatory_property(C,P), not assign(C,P,_).

12
13 :- assign(C,type ,T), assign(C,P,V), P != type , not property_val(T,P,V).

14 :- assign(C,type ,T), property_val(T,P,V), not assign(C,P,V).

15
16 :- require_com_com(C1 ,C2), component(C1), not component(C2).

17 :- require_com_pv(C1 ,(C2 ,P,V)), component(C1), not assign(C2 ,P,V).

18 :- require_pv_com ((C1 ,P,V),C2), assign(C1 ,P,V), not component(C2).

19 :- require_pv_pv ((C1 ,P1 ,V1),(C2 ,P2 ,V2)), assign(C1 ,P1 ,V1), not assign(C2 ,P2 ,V2).

20
21 :- incompatible_com_com(C1 ,C2), component(C1), component(C2).

22 :- incompatible_com_pv(C1 ,(C2 ,P,V)), component(C1), assign(C2 ,P,V).

23 :- incompatible_pv_pv ((C1 ,P1 ,V1),(C2 ,P2 ,V2)), assign(C1 ,P1 ,V1), assign(C2 ,P2 ,V2).

24
25 :- user_com(req ,C), not component(C).

26 :- user_com_pv(req ,(C,P,V)), not assign(C,P,V).

27 :- user_com(nreq ,C), component(C).

28 :- user_com_pv(nreq ,(C,P,V)), assign(C,P,V).

Figure 1: Bike Configuration in UML

4.2 Encoding and Solution

The encoding in Listing 2 is domain-independent and is separated into three parts: preprocessing in Lines
1-4, generation in Line 6-7, and testing in the remaining part.

In Line 1, the domains of the component properties are generated from the predefined property values
of their types. Assignment of values property of the components from their domain is given by Line 6, in
form of a choice rule. Also, to keep track of the components that are present in the configuration solution,
the predicate component(C) is used in Line 7, where C is the component.

The constraint A4, that all mandatory properties should be assigned a value if the component is
present, is represented in Line 11. Along with this, the constraint A1 is expressed using Line 2, where the
type of a component is set as a mandatory property.

Constraint A2 is represented in Line 9, meaning that a component property cannot have more than one
value. Line 13 and 14 depict rule A3. Line 13 expresses that a component cannot have property values
that the assigned type doesn’t have. In addition, Line 14 expresses that the component must have all the
pre-defined property values of its assigned type.

The partonomy constraints P1 and P2 are mapped into requirement constraint R1, expressed in Line
3 and 4. Line 3 expresses that the part of a whole component requires the whole component. In addition,
whole components require all their mandatory components to be present is enforced in Line 4. The
component requirements are defined in Line 16-19. Line 16 states that a component can require another
component (R1). Rule R2 is represented by Lines 17 and 18 where a component can require a certain
property value from another component and vice versa. The last requirement case, that a component with
a specific property value requires another component with a specific property value is represented by Line

19 (rule R3).
The incompatibility constraints are enforced in Lines 21-23. Line 28 translates constraint I1, where a

component is incompatible with another component. I2 is represented by Line 30. The last case that a
component with a certain property value is incompatible with another component with a specific property
value is done in Line 32 (I3).

The user requirements are represented in Line 25-28. A user can require a certain component (U1) or
a component with a specific property value (U2). This is represented in Line 25 and 26 respectively. On
the contrary, the user can also deny certain components or attribute values to be in the configuration (U3,
U4) which is represented in Line 27 and 28.

Each predicate in the solution assign(C,A,V) is intended to express that the component C with
attribute A and value V is present in the solution. Given the facts in Listing 1 and encoding in Listing 2, the
solution is represented in Listing 3. Since the user needs a basket that is incompatible with the mountain
bike, the user is assigned a city bike. Due to requirements in Lines 24 and 25 in Listing 1, stand is added
to the configuration, and f1 is selected as the frame. The wheel w2 is used as front wheel because of user
requirement. Due to incompatibility expressed in Line 21 of Listing 1, w2 is also used as the rear wheel.

Listing 3: Bike Configuration Solution
1 assign(bike ,type ,city_bike ). assign(rear_wheel ,type ,w2).

2 assign(front_wheel ,type ,w2). assign(frame ,type ,f1).

3 assign(stand ,type ,s1). assign(front_wheel ,material ,aluminium ).

4 assign(frame ,basket_support ,true). assign(basket ,type ,b1).

5 assign(frame ,material ,aluminium ). assign(front_wheel ,size ,26).

6 assign(rear_wheel ,size ,26). assign(rear_wheel ,material ,aluminium)

5 Future Work

This is a preliminary work with the final objective of developing an interactive product configuration system
[1] [10]. Identification and encoding of more concepts such as resource constraints, default values, user
preferences among others would be the next step. Mapping from commonly used graphical and constraint
based configuration representation methods to ASP should also be done. Interactive configuration has
many construction zones [25], including product recommendation, configuration diagnosis [12] and
explanation. Further investigation of these techniques in the context of ASP is another objective. Finally,
a major step would be implementation of these interactive techniques using clingo [7] along with its
sophisticated Python API that allows a fine-grained handling of the solving process.

